
Stataで簡単に試せるスプライン補間

第三回 重回帰・ロジスティック・Coxハザードでの
スプライン曲線の作図

mksplineで変数を変換してスプライン回帰をおこなう際，変換した変数以外にも独立変数をモデ

ルに投入したいわゆる重回帰をおこなうと，第二回までにお伝えした内容ではスプライン曲線の作

図が容易ではありません．本書では，xblcというユーザ作成の adoファイルを用いて，この重回帰
のスプライン曲線を作図する方法をお伝えします．

xblcのインストール

Stataを立ち上げ，コマンドウィンドウから以下を実行してください．

search xblc

searchコマンドでいくつかの主要な Stata関連サイトを検索し，xblcコマンドを見つけます．成

功すると，次のような画面が表示されます．

search for xblc (manual: [R] search)

Search of official help files, FAQs, Examples, SJs, and STBs

SJ-11-3 st0215_1 . Tab. and plot results after flex. modeling of quant. cov.
(help xblc if installed) . . . . . . . . . N. Orsini and S. Greenland
Q3/11 SJ 11(3):478
bug fix and added graph options for xblc

SJ-11-1 st0215 Tabulate and plot results after flex. modeling of quant. cov.
(help xblc if installed) . . . . . . . . . N. Orsini and S. Greenland
Q1/11 SJ 11(1):1--29
provides a postestimation command that facilitates the
presentation of the association between a quantitative
covariate and the response variable

st0215 1をクリックし，さらに (click here to install)をクリックするとダウンロードおよ

びインストールを開始します．処理は 10秒以下で完了することと思います．次のような表示が出れ
ば成功です．

installation complete.

(click here to return to the previous screen)

正常にインストールできた方は，次のコマンドで xblcのバージョンを確認します．

. which xblc
c:\ado\plus\x\xblc.ado
*! N.Orsini v.1.0.1 12jul11
*! N.Orsini v.1.0.0 19apr10

例題データ

例として用いるデータは第一回，第二回と同じものです．

webuse mksp1, clear

• 変数：

– id: 調査への参加者の ID

– age: 参加者の年齢

– educ: 教育期間（年）

– income: 所得（米ドル/年）

– lninc: incomeの対数

lnincを従属変数，ageを独立変数とする制約 3次スプライン回帰は，以下で実施できます．

mkspline agesp=age, cubic nknot(4)
regress lninc agesp*

スプライン曲線の作図は，以下のようなコマンドを doファイルに記述して実行します．

predict lninchat, xb
sort age
twoway line lninchat age, clcolor(navy) clwidth(thick) ///
|| scatter lninc age, mcolor(navy) ///
ytitle("ln(income)") ///
legend(label(1 "Restricted cubic fitted") ///
label(2 "Observed values")) ///
xline(16 35.35 48 68) xlabel(, grid)

9.
5

10
10

.5
11

1
1.

5
ln

(in
co

m
e)

20 40 60 80
age

Restricted cubic fitted Observed values

2



予測値の信頼区間の計算

xblcでは，信頼区間を次のように計算します．まず，従属変数を Y，mksplineで展開した後の独

立変数をX1，X2，X3と，mksplineで展開しなかった独立変数を Z4，Z5として，重回帰のモデル

式を次のように表すことにします．

Y = α + β1X1 + β2X2 + β3X3 + β4Z4 + β5Z5 + u (1)

mksplineで展開しなかった独立変数 Z4，Z5 をその平均値 Z̄4，Z̄5 で差し引き，いわゆるセンタ

リング（中心化）という変数変換を行って，Z4c，Z5c を便宜的に求めます．

Z4c = Z4 − Z̄4 (2)

Z5c = Z5 − Z̄5

Z4，Z5 の代わりに Z4c，Z5c を用いると，モデル式は次のようになります．

Y = α∗ + β1X1 + β2X2 + β3X3 + β4Z4c + β5Z5c + u (3)

α∗ = α − β4Z̄4 − β5Z̄5

この係数 α∗，β1，β2，β3，β4，β5について，重回帰による推定結果を α̂∗，β̂1，β̂2，β̂3，β̂4，β̂5

とします．

Y = α̂∗ + β̂1X1 + β̂2X2 + β̂3X3 + β̂4Z4c + β̂5Z5c + u (4)

それぞれの独立変数が，特定の値 x1，x2，x3，z4，z5 を取ったとき，予測値 ŷは次で書けます．

ŷ = α̂∗ + β̂1x1 + β̂2x2 + β̂3x3 + β̂4z4c + β̂5z5c (5)

誤差分布に正規分布を仮定し，α̂∗，β̂1，β̂2，β̂3，β̂4，β̂5が独立の分布に従うとすると，ŷの分散

V (ŷ)について，次のようにできます．

V (ŷ) = V
(
α̂∗ + β̂1x1 + β̂2x2 + β̂3x3 + β̂4z4 + β̂5z5

)
= V (α̂∗) + V

(
β̂1x1

)
+ V

(
β̂2x2

)
+ V

(
β̂3x3

)
+ V

(
β̂4z4

)
+ V

(
β̂5z5

)
= V (α̂∗) + x1V

(
β̂1

)
+ x2V

(
β̂2

)
+ x3V

(
β̂3

)
+ z4V

(
β̂4

)
+ z5V

(
β̂5

)
とくに z4c，z5cがゼロのとき，式 (2)より Z4，Z5それぞれが平均値をとるときの V (ŷ)を求める
ことになり，次のように簡単になります．

V (ŷ) = V (α̂) + x1V
(
β̂1

)
+ x2V

(
β̂2

)
+ x3V

(
β̂3

)
(6)

ŷの 95% 信頼区間は次のようになります．

ŷ ± z0.025

√
V (ŷ) = ŷ ± z0.025

√
V (α̂) + x1V

(
β̂1

)
+ x2V

(
β̂2

)
+ x3V

(
β̂3

)
X1，X2，X3について，mksplineで展開する前の変数をX とすると，ある値X = cに対応する

X1，X2，X3それぞれの値 c1，c2，c3を x1，x2，x3に選べば，スプライン曲線の横軸X について，

X = cでの信頼区間が上式から求まります．上式の計算について，Stataでは lincomを用いて簡単

に結果を求めることができます．

重回帰分析でのスプライン曲線の作図

まず，重回帰分析を実施します．新たに educを独立変数に加え，センタリングを行ったうえで回

帰をおこないます1．

sum educ

generate educ_c = educ - r(mean)

regress lninc agesp* educ_c

次に，xblcを利用してスプライン曲線を作図したいのですが，xblcではスプライン曲線の横軸に

する ageの各観測値をコマンド内で指定する必要があります．このため，levelsofという Stataに
標準搭載のコマンドを使用して各観測点の指定を手短かにおこないます．

. levelsof age
15 16 17 19 20 21 22 23 24 25 27 28 29 30 31 32 33 34 35 36 37 38
> 40 41 42 43 44 45 46 48 50 51 52 53 54 55 56 57 58 59 61 62 64
> 65 66 67 68 69

表示された各観測値は，r(levels)というマクロ変数に格納されています．この状態で xblcを実

行します．

xblc agesp*, covname(age) at(`r(levels)') line

上記コマンドについて詳述します，コマンド名 xblcに続き，regressで使用した変数 agesp*を入

力します．カンマ（,）を入力し，covname()オプションを入力して，カッコ内に本来の観測値である

変数 ageを入力します．at()オプションを入力し，ageの各観測値を入力するのですが，r(levels)

に格納されているのでシングルクォーテーションでくくって at()のカッコ内に記入します．ここで用

いるシングルクォーテーションは Stataの特別なルールに従った記号を使用します．シングルクォー
テーションの開始は Shift + @キーで入力できる記号を使用します．シングルクォーテーションの

終了は Shift + 7キーで入力できる記号を使用します．lineオプションを入力するとスプライン曲

線が作図されます．
1センタリングをするのは，=0 で平均値を示すようにするためです．スプライン曲線で２軸に表さない独立変数（または

共変量）は，連続変数の場合その値がゼロのとき，またカテゴリ変数の場合その値が最小値のときのグラフがそれぞれ描かれ
ます．すなわち，連続変数については，センタリング後の変数を用いることで，センタリング前の変数の平均値でのグラフを
描くことができます．同様に，カテゴリ変数については、推定コマンド（logit など) で ib#. を用いて基準（ベース）を指
定することで，その値でのグラフを描くことができます．

levelsofコマンドで指定する変数が小数を含み，且つ double型（Stataの倍精度型）の場合，そ
の後の xblc で値が存在しないという旨のエラーが起きてしまいます．おそらく xblc コマンドが

float型（Stataの単精度型）にしか対応していないことが原因だと考えられます．この場合，double
型で保存されている値を別の変数に退避させ，元の変数のほうは float型に型変換をおこなってから
levelsofおよび xblcを実行すると成功します．

generate age_double = age

recast float age, force

levelsof age

xblc agesp*, covname(age) at(`r(levels)') line

ロジスティック回帰でのスプライン曲線の作図

ロジスティック回帰ではオッズ比のスプライン曲線を作図することになると思います．オッズ比の

プロット時は xblcコマンドで eformオプションを入力します．オッズ比は基準のオッズが必要にな

りますので，reference()オプションで基準となる観測点を指定します．基準となる観測点はデー

タセット内に存在している必要があります．横軸の値をカッコ内に入力して指定します2．

use http://nicolaorsini.altervista.org/data/pa_luts, clear

quietly summarize age

generate agec = age - r(mean)

mkspline tpas = tpa, knots(37.2 39.6 42.3 45.6) cubic

logit ipss2 tpas1 tpas2 tpas3 agec

levelsof tpa

xblc tpas1-tpas3, covname(tpa) at(`r(levels)') reference(29) eform line

Cox比例ハザードモデルでのスプライン曲線の作図

Cox比例ハザードモデルでは，前述のロジスティック回帰と同様の方法で作図できます．通常は，
ハザード比のスプライン曲線を作図することになると思います．ハザード比のプロット時は xblc

コマンドで eform オプションを入力します．ハザード比は基準のハザードが必要になりますので，

reference()オプションで基準となる観測点を指定します．基準となる観測点はデータセット内に

存在している必要があります．横軸の値をカッコ内に入力して指定します3．

webuse stan3, clear

stset

2以下の例は xblc のヘルプにある Example のコマンドを少し修正したものです
3以下の例は stcox のヘルプにある Example のコマンドを少し修正したものです

mkspline ages = age, nknot(3) cubic

quietly sum year

generate year_c = year - r(mean)

stcox ages* year_c

levelsof age

xblc ages*, covname(age) at(`r(levels)') reference(30) eform line

